

Las funciones que trae PSeInt.

No es el momento de explicar funciones en programación, más adelante se verá en más de una clase, pero si es bueno ir conociendo las que tiene definidas nuestro PSeInt.

Para usar las funciones se debe colocar su nombre y entre paréntesis el o los argumentos que requiera. Esta función devolverá un valor y nos facilitará realizar ciertos cálculos o usar código ya escrito.

Sintaxis: nombreFuncion (arg1, arg2, ..., argN)

Donde nombreFuncion es el nombre de la función y arg1, ... son los argumentos que requiere la función para devolvernos un resultado.

Funciones matemáticas

[-] Func. Matemáticas	
	abs (valor absoluto)
	trunc (valor truncado)
	redon (valor redondeado)
	raiz (raíz cuadrada)
	sen (seno)
	cos (coseno)
	tan (tangente)
	asen (arcoseno)
	acos (arcocoseno)
	atan (arcotangente)
	ln (logaritmo natural)
	exp (func. exponencial)
	azar (numero aleatorio)

Veamos Algunos ejemplos

Algoritmo funciones_numericas

Definir x **Como** Entero

x=-3

// abs (x) devuelve el valor absoluto de x

Escribir "Valor Absoluto: ",abs(x)

x=21

// azar (x) devuelve un numero al azar entre 0 y x-1

Escribir "Azar: ", azar(x)

// raiz(x) devuelve la raiz cuadrada de x

Escribir "Raiz cuadrada: ", raiz(x)

// redon(x) redondea x al entero mas cercano

Escribir "Redondeo: ", redon(1.34)," ::: ", redon(8.76)

// trunc(x) devuelve la parte entera de x

Escribir "Truncado: ", trunc(1.34)," ::: ", trunc(8.76)

FinAlgoritmo

*** Ejecución Iniciada. ***

Valor Absoluto: 3

Azar: 15

Raiz cuadrada: 4.582575695

Redondeo: 1 ::: 9

Truncado: 1 ::: 8

*** Ejecución Finalizada. ***

Funciones para cadenas de texto

[-] Func. p/Cadenas	
	Longitud
	SubCadena
	Concatenar
	ConvertirANumero
	ConvertirATexto
	Mayusculas
	Minusculas

Veamos Algunos ejemplos

Algoritmo funciones_cadenas

Definir x Como Caracter

x="Hola Mundo"

// Longitud (x) retorna la cantidad de caracteres de la cadena x.

Escribir Longitud(x)

// Concatenar(cad1,cad2) retorna una sola cadena con la union de ambas

Escribir Concatenar(x," lindo")

// Mayusculas(x) regresa la cadena en mayusculas

Escribir Mayusculas(x)

// Minusculas(x) regresa la cadena en minusculas

Escribir Minusculas(x)

// SubCadena(x,desde,hasta) extrae una cadena de x que

// comienza en desde y termina en hasta

Escribir SubCadena(x,2,6)

// ConvertirANumero(x) convierte a numero entero o real una

// cadena que contenga valores numericos

Escribir ConvertirANumero("34.3")

// ConvertirATexto(x) siendo x un numero, lo devuelve como cadena de texto

Escribir ConvertirATexto(33.4)

FinAlgoritmo

*** Ejecución Iniciada. ***

10

Hola Mundo lindo

HOLA MUNDO

hola mundo

ola M

34.3

33.4

*** Ejecución Finalizada. ***